

TD-TP 14 proba version 1

TD et TP à faire en 2 heures seulement . **Attention !**

Concepts : type dérivés, type numériques flottants prédéfinis pour cette séance (`Float`), surcharge, privatisation et type abstrait de données encore !

Conseils : On utilisera uniquement pour ce petit exercice le type prédéfini `Float` à l'exclusion de tout autre. Pour éditer proprement un `Float` on utilisera ou on copiera la procédure `Ecrire` du vieux paquetage `P_E_Sortie`. Suivez les conseils ci-dessous.

Thème : application du cours arithmétique aux concepts de Probabilités (connus?) des étudiants. Création d'un type abstrait de données associé `T_Proba`. Cet exercice (TD en une heure) est approfondi dans le TP (en une heure aussi) qui suit.

Rappel (math): si A et B dénotent des évènements (**indépendants**) alors on a:

$$\begin{aligned}\Pr(A \text{ ou } B) &= \Pr(A) + \Pr(B) - \Pr(A \text{ et } B) \\ \Pr(A \text{ et } B) &= \Pr(A) * \Pr(B)\end{aligned}$$

Sinon quand A et B ne sont pas indépendants on a :

$$\Pr(A \text{ et } B) = \Pr(A) * \Pr(B_{/A}) \text{ ou encore } \Pr(A \text{ et } B) = \Pr(B) * \Pr(A_{/B})$$

Remarque on notera `Pr` l'opérateur de Probabilité (ancienne notation !) alors que actuellement (revoir vos cours du BAC) c'est `P` !

Exemple : `A` \equiv être une fille et `B` \equiv avoir les cheveux blonds. Alors on dit que A et B sont deux évènements indépendants et :

Si $\Pr(\text{être une fille}) = 0.4 = \Pr(A)$ et si $\Pr(\text{avoir les cheveux blonds}) = 0.3 = \Pr(B)$ alors on a :

$$\begin{aligned}\Pr(\text{être fille ou être blond(e)}) &= 0.4 + 0.3 - 0.12 = 0.58 = \Pr(A \text{ ou } B) \\ \Pr(\text{être fille et être blonde}) &= 0.4 * 0.3 = 0.12 = \Pr(A \text{ et } B)\end{aligned}$$

On va noter `a` = `Pr (A)` et `b` = `Pr (B)`. Comprenez bien ces écritures raccourcies !

On définit ainsi un nouvel ensemble de nombres (`a`, `b`,...) qui prend ses valeurs dans l'intervalle `0.0..1.0` (ce sont les Probas). Cet ensemble sera un type à part (**T.A.D.**). Dont on va définir la structure et les méthodes.

On utilisera pour noter $\Pr(A \text{ ou } B)$ l'expression: $a "+" b$ (**ceci n'est qu'une convention**).

Pour réécrire $\Pr(A \text{ ou } B)$ ou $a "+" b$ on utilise la définition suivante :

$$a "+" b = a + b - a * b \quad (1)$$

noter le "+" et le + (ils n'agissent pas sur les mêmes objets et risquent d'induire des problèmes). En effet dans le membre de gauche il s'agit de la nouvelle notation (dénotee "+"). Dans le membre de droite il s'agit d'opérations sur des objets pris au sens de valeurs numériques réelles. On voit qu'il faudra bien différencier les objets de type `T_Proba` d'une part, des numériques réels (`Float`) d'autre part. Et pour ce faire on utilisera `private` pour que l'utilisateur n'utilise que les opérations (ou méthodes) dédiées au type `T_Proba`. A l'intérieur de la partie privée on implémentera les `T_Proba` avec des `Float`. On pourra dériver le type `Float` pour réaliser `T_Proba`..

On désire réaliser en Ada cette surcharge "+" et plus généralement le corps du paquetage suivant :

```
package P_Proba1 is -- exercice ecrire le body correspondant
  type T_Proba is private;
  function "+" (Un,Deux : in T_Proba) return T_Proba;
  function "*" (Un,Deux : in T_Proba) return T_Proba;
  function "-" (Un : in T_Proba) return T_Proba;

  procedure Get (Un : out T_Proba); -- lecture validée
  procedure Put (Un : in T_Proba); -- édition agréable
  procedure Get (Fic : in File_Type ; Un : out T_Proba) ;
  procedure Put (Fic : in File_Type; Un : in T_Proba);

private
  type T_Proba is new Float range 0.0..1.0;
end P_Proba1;
```

Aides :

Si on écrit (dans le body) ceci (trop hâtivement !) :

```
function "+" (X,Y: in T_Proba) return T_Proba is
begin
  return X + Y - X * Y ; -- (2)
end "+";
```

La surcharge "+" pose un problème en effet :

Dans le membre de droite de (1) ($a "+" b = a + b - a*b$) que signifie $a + b$? c'est le + habituel agissant sur la représentation numérique de a et b et non pas une **définition récursive** de la surcharge comme, mise par erreur, dans (2) corps de la fonction. L'écriture (2) ci-dessus **ne réalise donc pas les spécifications** de (1).

Il faut **convertir les opérandes** de l'expression $X + Y - X * Y$ dans le type père (Float ici).

La fonction "-" (unaire) des spécifications est tout simplement la Probabilité de l'évènement contraire c'est-à-dire : $-(A) = 1 - A$.

Vous testerez votre body avec le programme de test suivant (minimum). Rappel : la lecture clavier Get doit être validée proprement (dans le body), l'écriture (Put) doit être de qualité (cf. Ecrire d'un Float de P_E_Sortie).

Au moment du TP d'autres tests seront ajoutés (dans le fichier) .

```
with Ada.Text_IO;
use Ada.Text_IO;
with P_Probal;
use P_Probal;

procedure Pp_Probl is

    Fic_In,
    Fic_Out : File_Type;
    A,
    B      : T_Proba;

begin
    Open (Fic_In, In_File, "proba.in");
    Create (Fic_Out, Name => "proba.out");
    loop
        exit when End_Of_File (Fic_In);
        Get (Fic_In, A);
        Get (Fic_In, B);
        Skip_Line(Fic_In);
        Put (Fic_Out, "probabilite d'être une femme: ");
        Put (Fic_Out, A);
        New_Line (Fic_Out);
        Put (Fic_Out, "probabilite d'avoir les cheveux blonds: ");
        Put (Fic_Out, B);
        New_Line (Fic_Out);
        Put (Fic_Out, "probabilite d'être femme ou d'etre blond(e): ");
        Put (Fic_Out, A + B);
        New_Line (Fic_Out);
        Put (Fic_Out, "probabilite d'être une femme blonde: ");
        Put (Fic_Out, A * B);
        New_Line (Fic_Out);
        Put (Fic_Out, "probabilite d'être un homme: ");
        Put (Fic_Out, - A);
        New_Line (Fic_Out);
    end loop;
    Close (Fic_In);
    Close (Fic_Out);
end Pp_Probl;
```

Détails du TP :

- Créez un répertoire TP14.
- Copiez les fichiers proposés.
- Réalisez rapidement le body de P_Proba1 à l'aide du squelette p_proba1.adb.squ à renommer.
- Faites plusieurs essais avec pp_prob1 (en complétant le fichier de données).
- Editez le fichier des résultats et le body

Ceci est le premier travail de 2 heures TD et TP. Ce travail est utile pour le TD-TP suivant dit version 2. **Soignez la qualité de ce TP** version 1 qui conditionne la qualité et la rapidité du second, de 2 heures aussi, et qui récupère beaucoup de celui-ci !